
Software for multiplier ideals

ABSTRACT: We describe a new software package for computing multiplier
ideals in certain cases, including monomial ideals, monomial curves, generic de-
terminantal ideals, and hyperplane arrangements. In these cases we take advan-
tage of combinatorial formulas for multiplier ideals given by results of Howald,
Thompson, and Johnson. The package uses the package Normaliz. It is available
as a library for Macaulay2.

INTRODUCTION. Multiplier ideals have been applied to a number of problems in
algebraic geometry in recent years, most spectacularly in recent major advances
in the minimal model program [Hacon and McKernan 2007; Birkar et al. 2010]
that built on earlier work showing the deformation invariance of plurigenera [Siu
1998]. Other applications include several results on singularities and linear series
[Lazarsfeld 2004; Ein and Mustat,ă 2006], a bound for symbolic powers [Ein et al.
2001], and applications to algebraic statistics [Watanabe 2009; Zwiernik 2011;
Drton et al. 2009, Chapter 5]. New applications of multiplier ideals continue to
emerge in topics such as Chow stability [Lee 2008] and singularities in generic
liaison [Niu 2014]. With broad and growing interest in multiplier ideals, it is
increasingly valuable to compute examples.

For a thorough introduction to multiplier ideals see [Lazarsfeld 2004]. Here is a
definition of multiplier ideals in terms of resolution of singularities: Suppose X is
a smooth variety over a field k (we may assume X is affine, or even just kn , since
we are primarily interested in local issues), I ⊂ OX is a nonzero ideal sheaf, and
µ : Y → X is a log resolution of I , so that the total transform I OY defines a divisor
F with simple normal crossings support, F =

∑
ai Ei , where the Ei are distinct

reduced components of F . Then for each real number c ≥ 0 the c-th multiplier
ideal is defined by

J(I c)= µ∗OY (KY/X −bc · Fc),

where KY/X is the relative canonical divisor of Y over X , defined locally by the
vanishing of the determinant of the Jacobian dµ, and bc ·Fc denotes the component-
wise round-down of the R-divisor c · F , given by bc · Fc =

∑
bcaicEi .

MSC2010: primary 14Q99, 14F18; secondary 13A15, 13P25.
Keywords: multiplier ideal, log canonical threshold, jumping number.

In theory it is algorithmic to compute multiplier ideals by computing a resolution
of singularities of I followed by a sheaf pushforward. In practice it is more difficult;
see [Frühbis-Krüger 2014].

Shibuta’s [2011] algorithm for computing Bernstein–Sato polynomials and mul-
tiplier ideals via Gröbner basis methods in Weyl algebras (which he implemented
in Risa/Asir) was refined and implemented in the Dmodules library for Macaulay2
by Berkesch and Leykin [2010]. The Dmodules library can compute multiplier
ideals and jumping numbers of arbitrary ideals, but due to the difficulty of the
computations can only handle modestly sized examples.

We describe a new software package named MultiplierIdeals that computes
multiplier ideals of special ideals, including monomial ideals, ideals of monomial
curves, generic determinantal ideals, and hyperplane arrangements, via combina-
torial methods, using the Normaliz software and interface to Macaulay2 by Bruns,
Ichim, and Kämpf [Bruns and Ichim 2010; Bruns and Kämpf 2010]. The com-
binatorial methods allow computations of somewhat larger examples than can be
handled by general methods.

Wherever possible we work over an arbitrary field k. Since multiplier ideals
in our cases are computed by resolutions defined over Z (or over the Z-algebra
generated by the coefficients of the defining equations of the input data), we may
work in arbitrary characteristic.

Our package also computes certain quantities associated to multiplier ideals:
the log canonical thresholds and jumping numbers. Because of the round-down
operation, J(I c+ε) = J(I c) for sufficiently small ε > 0. A real number c ≥ 0 is
a jumping number of I if J(I c) 6= J(I c−ε) for all ε > 0. Every jumping number
is in fact rational. The smallest strictly positive jumping number is called the log
canonical threshold of I , denoted lct(I ). It turns out that J(I 0)= (1) is the trivial
ideal, so lct(I ) is the supremum of c such that J(I c)= (1); equivalently, lct(I ) is
the first value of c such that J(I c) 6= (1).

The portion dealing with monomial ideals was written first and distributed as
the package MonomialMultiplierIdeals. The portion dealing with monomial curves
was written by C. Raicu, B. Snapp, and the author at the 2011 IMA Special Work-
shop on Macaulay2, and distributed as the package SpaceCurvesMultiplierIdeals.
The portion dealing with hyperplane arrangements is based on code written by
Denham and Smith for the HyperplaneArrangements package [2011]. These por-
tions were all integrated into the present package, and computations with generic
determinantal ideals added, at the 2012 Macaulay2 Workshop at Wake Forest.

MONOMIAL IDEALS. For a monomial ideal I ⊂ k[x1, . . . , xn], let monom(I )⊂
Zn
≥0 be the set of exponent vectors of monomials in I . The Newton polyhedron

Newt(I ) is the convex hull of monom(I ). Let 1= (1, . . . , 1) ∈ Rn .

Multiplier ideals of monomial ideals are described by the following theorem
of Howald:

Theorem 1 [Howald 2001; Blickle 2004]. The multiplier ideal J(I c) is the mono-
mial ideal containing xv if and only if v+ 1 ∈ Int(c ·Newt(I )). Here Int denotes
the topological interior of c ·Newt(I ) relative to the nonnegative orthant, that is,
as a subset of (R≥0)

n .

In other words, the multiplier ideal is the quotient ideal

J(I c)=
(
xv : v ∈ Int(c ·Newt(I ))

)
: x1.

The Newton polyhedron Newt(I ) is defined by a system of inequalities Av ≥ b,
where A is an r × n matrix, b is a vector, and ≥ is the partial order of entry-wise
comparison, where a ≥ b if and only if ai ≥ bi , 1 ≤ i ≤ r . Then c ·Newt(I ) is
defined by Av ≥ cb. The interior Int(c ·Newt(I )) is the solution of the system of
inequalities given by {

Aiv > cbi if bi 6= 0,
Aiv ≥ cbi = 0 if bi = 0.

Since Newt(I ) is a rational polyhedron, we can (and do) take the A and b to
have integer entries. Furthermore, since I is an ideal the entries of A and b are
nonnegative. In practice, it is sufficient to compute J(I c) for rational c = p/q , and
this can be done as follows: To find the integer vectors v lying in the topological
interior of the solution region to Av ≥ cb (equivalently, q Av ≥ pb), we add 1 to
the nonzero entries of pb, yielding a vector b′ with entries b′i = pbi + 1 if bi 6= 0,
and b′i = pbi = bi = 0 otherwise. Then the multiplier ideal J(I c) is the quotient
(xv : q Av ≥ b′) : x1.

The software Normaliz can compute the defining inequalities Av≥ b of Newt(I )
and the solutions to the modified system q Av ≥ b′; Macaulay2 can compute the
ideal quotient by x1, giving the multiplier ideal:

Macaulay2, version 1.6
i1 : needsPackage "MultiplierIdeals";
i2 : R = QQ[x,y,z,w];
i3 : I = monomialIdeal(x*y, x*z, y*z, y*w, z*w^2);
o3 : MonomialIdeal of R
i4 : logCanonicalThreshold(I)
o4 = 2
i5 : multiplierIdeal(I,7/3)

2
o5 = ideal (y, z*w, z , x*z)
i6 : toString jumpingNumbers(I)
o6 = {{2, 7/3, 5/2, 8/3, 3, 10/3, 7/2, 11/3, 4}, {ideal(z,y),

ideal(y,z*w,z^2,x*z), ideal(z*w,y*w,y*z,x*z,y^2,x*y),
ideal(y*w,y*z,x*z,y^2,x*y,z*w^2,z^2*w),
ideal(y*z*w,y^2*w,y*z^2,x*z^2,y^2*z,x*y*z,x*y^2,z^2*w^2), ...

The jumpingNumbers command produces a list with two elements:

(1) A list of the jumping numbers of I in the interval (0, k(I )], where k(I ) is
the analytic spread of I . A different interval may be specified as an optional
argument.

(2) A list of the multiplier ideals at the jumping numbers. (The list is truncated
in the above example.)

Thus the output of the last command says that this ideal I has jumping num-
bers 2, 7/3, . . . , and gives the corresponding multiplier ideals: J(I 2) = (z, y),
J(I 7/3) = (y, zw, z2, xz), and so on. Multiplier ideals and jumping numbers for
c> k(I ) are given by Skoda’s theorem [Lazarsfeld 2004, Theorem 9.6.21]. Namely,
for c > k(I ), J(I c)= I J(I c−1); and c > k(I ) is a jumping number if and only if
c− 1 is a jumping number.

In the above example, the log canonical threshold, single multiplier ideal J(I 7/3),
and list of nine jumping numbers and multiplier ideals were each computed in a
fraction of a second on a 2012 MacBook with dual-core 64-bit 2.9 GHz CPU and
8 GB RAM. By way of comparison, the Dmodules package takes about 42 seconds
to compute the log canonical threshold on the same machine, and about 84 seconds
to compute J(I 7/3). This comparison is only intended to illustrate the advantages
of using special algorithms where available, and we remind the reader that the
DModules package uses a general method.

For monomial ideals, extra information is available: for any monomial xv, the
package computes the threshold value min{c : xv /∈ J(I c)}, and the list of facets of
the Newton polyhedron that impose the nonmembership:

i7 : toString logCanonicalThreshold(I,z^2*w)
o7 = (3,matrix {{2, 2, 1, 1, -3}, {2, 2, 0, 1, -2}})

This output means that z2w /∈ J(I 3) but z2w ∈ J(I c) for c < 3. That is, for the
exponent vector v = (0, 0, 2, 1), v + 1 lies on the boundary of 3 ·Newt(I ); and
furthermore it lies on the intersection of two facets, the ones scaled up from the
facets of Newt(I ) defined by 2x + 2y+ z+w = 3 and 2x + 2y+w = 2.

The log canonical threshold of the ideal I itself is the threshold value for 1= x0.

MONOMIAL CURVES. An affine monomial curve is one parametrized by t 7→
(ta1, . . . , tan ). We can and do assume that 1≤a1≤· · ·≤an and gcd(a1, . . . , an)=1.
For convenience we denote this curve by C(a1, . . . , an). It has a singularity at the
origin when a1≥ 2. The defining ideal is the kernel of the map k[x1, . . . , xn]→k[t]
given by xi 7→ tai . This is a binomial ideal.

The multiplier ideals of affine monomial curves in dimension n = 3 have been
found by Howard Thompson [2014],1 using the combinatorial description of the

1This paper states the result over C, but it holds over any field.

resolution of singularities of a binomial ideal given in [González Pérez and Teissier
2002]. This yields a combinatorial formula in terms of the vector (a, b, c) of ex-
ponents appearing in the parametrization t 7→ (ta, tb, tc). Our software package
implements Thompson’s result, again calling on Normaliz to find generators for
the semigroup of integer solutions to certain linear inequalities:

i8 : R = QQ[x,y,z]; S = QQ[t];
i10 : I = kernel map(S,R,{t^3,t^4,t^5}) -- ideal of C(3,4,5)

2 2 2 3
o10 = ideal (y - x*z, x y - z , x - y*z)

To compute the multiplier ideals and log canonical threshold of I , we input the list
of exponents in the parametrization:

i11 : toString logCanonicalThreshold(R,{3,4,5})
o11 = 13/9
i12 : multiplierIdeal(R,{3,4,5},13/9)
o12 = ideal (z, y, x)

GENERIC DETERMINANTAL IDEALS. Let X = (xi, j )1≤i≤m,1≤ j≤n be an m × n
generic matrix, meaning one whose entries are independent variables. Let Ir (X)
be the ideal generated by the r × r minors of X . The multiplier ideals of Ir (X)
have been found by Amanda Johnson [2003]:2

Theorem 2. With X , m, n, and r as above, the multiplier ideals are given by the
following intersection of symbolic powers of determinantal ideals:

J(Ir (X)c)=
r⋂

i=1

Ii (X)(bc(r+1−i)c+1−(n−i+1)(m−i+1)).

Recall that symbolic powers of generic determinantal ideals may be expressed as

Ir (X)(a) =
∑

κ1+···+κs=a

s∏
i=1

Ir−1+κi (X),

with the sum taken over partitions of a. See [Bruns and Vetter 1988, Theorem 10.4].
We may compute multiplier ideals of determinantal ideals in our software by

giving the matrix X and the size of minors. Here we examine multiplier ideals of
the size-2 and size-3 minors of a 4× 5 generic matrix:

i13 : x = getSymbol"x"; R = QQ[x_1..x_20];
i15 : X = genericMatrix(R,4,5); -- a 4x5 generic matrix
i16 : logCanonicalThreshold(X,2) -- lct of the ideal of 2x2 minors
o16 = 10
i17 : multiplierIdeal(X,2,10) == minors(1,X) -- J(I^10) where I = 2x2 minors
o17 = true

2This dissertation states the result for algebraically closed fields, but it holds over any field.

i18 : multiplierIdeal(X,2,11) == (minors(1,X))^3 -- J(I^11)
o18 = true

HYPERPLANE ARRANGEMENTS. A formula for multiplier ideals of hyperplane
arrangements was found by Mustat,ă [2006] and simplified in [Teitler 2008].3 The
HyperplaneArrangements package [Denham and Smith 2011] uses these results
to compute multiplier ideals and log canonical thresholds of hyperplane arrange-
ments. To this we add the ability to compute jumping numbers and other minor
modifications. I thank Graham Denham and Gregory G. Smith, the authors of
HyperplaneArrangements, for their permission to copy and modify their package’s
source code.

The following is Example 6.3 of [Berkesch and Leykin 2010]:
i19 : R = QQ[x,y,z];
i20 : ff = toList factor ( (x^2-y^2)*(x^2-z^2)*(y^2-z^2)*z ) / first;
i21 : A = arrangement ff;
i22 : toString jumpingNumbers(A,IntervalType=>"ClosedOpen")
o22 = {{3/7, 4/7, 2/3, 6/7}, {ideal(z,y,x), ideal(z^2,y*z,x*z,y^2,x*y,x^2),

ideal(y^2*z-z^3,x^2*z-z^3,x*y^2-x*z^2,x^2*y-y*z^2), ...

ACKNOWLEDGEMENTS. I am very grateful to Claudiu Raicu and Bart Snapp
for their critical contributions to the package and for a number of very helpful
comments about this paper and the software package itself. I would also like
to thank Howard Thompson for sharing his work-in-progress and for numerous
helpful conversations, Graham Denham and Gregory G. Smith, the organizers of
the 2011 IMA Special Workshop on Macaulay2, and the organizers of the 2012
Macaulay2 Workshop at Wake Forest.

REFERENCES.
[Berkesch and Leykin 2010] C. Berkesch and A. Leykin, “Algorithms for Bernstein–Sato polyno-
mials and multiplier ideals”, pp. 99–106 in Proceedings of the 2010 International Symposium on
Symbolic and Algebraic Computation, edited by S. M. Watt, ACM, New York, 2010. MR 2920542

[Birkar et al. 2010] C. Birkar, P. Cascini, C. D. Hacon, and J. McKernan, “Existence of minimal mod-
els for varieties of log general type”, J. Amer. Math. Soc. 23:2 (2010), 405–468. MR 2011f:14023
Zbl 1210.14019

[Blickle 2004] M. Blickle, “Multiplier ideals and modules on toric varieties”, Math. Z. 248:1 (2004),
113–121. MR 2006a:14082 Zbl 1061.14055

[Bruns and Ichim 2010] W. Bruns and B. Ichim, “Normaliz: algorithms for affine monoids and
rational cones”, J. Algebra 324:5 (2010), 1098–1113. MR 2011m:20001 Zbl 1203.13033

[Bruns and Kämpf 2010] W. Bruns and G. Kämpf, “A Macaulay2 interface for Normaliz”, J. Softw.
Algebra Geom. 2 (2010), 15–19. MR 2881130

[Bruns and Vetter 1988] W. Bruns and U. Vetter, Determinantal rings, Lecture Notes in Mathematics
1327, Springer, Berlin, 1988. MR 89i:13001 Zbl 0673.13006

3These papers state the result for C, but it holds over any field.

http://dx.doi.org/10.1145/1837934.1837958
http://dx.doi.org/10.1145/1837934.1837958
http://msp.org/idx/mr/2920542
http://dx.doi.org/10.1090/S0894-0347-09-00649-3
http://dx.doi.org/10.1090/S0894-0347-09-00649-3
http://msp.org/idx/mr/2011f:14023
http://msp.org/idx/zbl/1210.14019
http://dx.doi.org/10.1007/s00209-004-0655-y
http://msp.org/idx/mr/2006a:14082
http://msp.org/idx/zbl/1061.14055
http://dx.doi.org/10.1016/j.jalgebra.2010.01.031
http://dx.doi.org/10.1016/j.jalgebra.2010.01.031
http://msp.org/idx/mr/2011m:20001
http://msp.org/idx/zbl/1203.13033
http://dx.doi.org/10.2140/jsag.2010.2.15
http://msp.org/idx/mr/2881130
http://msp.org/idx/mr/89i:13001
http://msp.org/idx/zbl/0673.13006


Teitler :::: Software for multiplier ideals 7

[Denham and Smith 2011] G. Denham and G. G. Smith, “HyperplaneArrangements, version 0.9 —
a package for Macaulay2”, 2011, http://www.math.uiuc.edu/Macaulay2/doc/Macaulay2-1.6/share/
doc/Macaulay2/HyperplaneArrangements/html/.

[Drton et al. 2009] M. Drton, B. Sturmfels, and S. Sullivant, Lectures on algebraic statistics, Ober-
wolfach Seminars 39, Birkhäuser, Basel, 2009. MR 2012d:62004 Zbl 1166.13001

Partially ordered sets in Macaulay2

DAVID COOK II, SONJA MAPES AND GWYNETH WHIELDON

ABSTRACT: We introduce the package Posets for Macaulay2. This package
provides a data structure and the necessary methods for working with partially
ordered sets, also called posets. In particular, the package implements methods
to enumerate many commonly studied classes of posets, perform operations on
posets, and calculate various invariants associated to posets.

INTRODUCTION. A partial order is a binary relation � over a set P that is
antisymmetric, reflexive, and transitive. A set P together with a partial order �
is called a poset, or partially ordered set. We refer the reader to the seminal text
[Stanley 2012] for definitions omitted herein.

Posets are combinatorial structures that are used in modern mathematical research,
particularly in algebra. We introduce the package Posets for Macaulay2 [Grayson
and Stillman] via three distinct posets or related ideals which arise naturally in
combinatorial algebra.

We first describe two posets that are generated from algebraic objects. The
intersection semilattice associated to a hyperplane arrangement can be used to
compute the number of unbounded and bounded real regions cut out by a hyper-
plane arrangement, as well as the dimensions of the homologies of the complex
complement of a hyperplane arrangement.

Given a monomial ideal, the lcm-lattice of its minimal generators gives informa-
tion on the structure of the free resolution of the original ideal. Specifically, two
monomial ideals with isomorphic lcm-lattices have the “same” (up to relabeling)
minimal free resolution, and the lcm-lattice can be used to compute, among other
things, the multigraded Betti numbers

βi,b(R/M)= dimK Tori,b(R/M, K )

of the monomial ideal.
In contrast to the first two examples (associating a poset to an algebraic object),

we then describe an ideal that is generated from a poset. In particular, the Hibi ideal
of a finite poset is a squarefree monomial ideal which has many nice algebraic

MSC2010: 06A11, 13P99.
Keywords: posets.

p1

p2

p3 p4

H1 H2 H3

H4

0̂

H1 H2 H3 H4

p1 p4 p3 p2

Figure 1. Left: the noncentral hyperplane arrangement A comprising
H1= V (x+y), H2= V (x), H3= V (x−y), H4= V (y+1). Right: The
intersection meet semilattice L(A) of A.

properties that can be described in terms of combinatorial properties of the poset.
For example, the minimal free resolution, the Betti numbers, and the projective
dimension are nicely described in terms of data about the poset itself.

INTERSECTION (MEET SEMI)LATTICES. A hyperplane arrangement A is a finite
collection of affine hyperplanes in some vector space V . The dimension of a hyper-
plane arrangement is defined by dim(A)= dim(V ), and the rank of a hyperplane
arrangement rank(A) is the dimension of the span in V of the set of normals to the
hyperplanes in A.

The intersection (meet semi)lattice L(A) of A is the set of the nonempty inter-
sections of subsets of hyperplanes

⋂
H∈A′ H for H ∈ A′ ⊆ A, ordered by reverse

inclusion. We include the empty intersection corresponding to A′ =∅, which is the
minimal element 0̂= V in the intersection meet semilattice L(A). If the intersection
of all hyperplanes in A is nonempty, i.e.,

⋂
H∈A H 6=∅, then the intersection meet

semilattice L(A) is actually a lattice. Arrangements with this property are called
central arrangements.

Consider the noncentral hyperplane arrangement

A= {H1 = V (x + y),H2 = V (x),H3 = V (x − y),H4 = V (y+ 1)},

where Hi = V (`i (x, y))⊆ R2 denotes the hyperplane of zeros of the linear form
`i (x, y); see Figure 1, left. We construct L(A) in Macaulay2 as follows.

i1 : needsPackage "Posets";
i2 : R = RR[x,y];
i3 : A = {x + y, x, x - y, y + 1};
i4 : LA = intersectionLattice(A, R);

Further, using the method texPoset we generate LATEX code to display the Hasse
diagram of L(A), as in Figure 1, right.

A theorem of Zaslavsky [1975] provides information about the topology of the
complement of hyperplane arrangements in Rn . Let µ denote the Möbius function
of the intersection meet semilattice L(A). Then the number of regions that A
divides Rn into is

r(A)=
∑

x∈L(A)

|µ(0̂, x)|.

Moreover, the number of these regions that are bounded is

b(A)= |µ(L(A)∪ 1̂)|,

where L(A)∪1̂ is the intersection meet semilattice adjoined with a maximal element.

We verify these results for the noncentral hyperplane arrangement A using
Macaulay2:

i5 : realRegions(A, R)
o5 = 10
i6 : boundedRegions(A, R)
o6 = 2

Moreover, in the case of hyperplane arrangements in Cn , using a theorem of
Orlik and Solomon [1980] we can recover the Betti numbers (dimensions of ho-
mologies) of the complement MA = Cn

−
⋃

A of the hyperplane arrangement
using purely combinatorial data of the intersection meet semilattice. In particular,
MA has torsion-free integral homology with Betti numbers given by

βi (MA)= dimC(Hi (MA))=
∑

x∈L(A)
dimC(x)=n−i

|µ(0̂, x)|,

where µ( · ) again represents the Möbius function. See [Wachs 2007] for details
and generalizations of this formula.

Posets will compute the ranks of elements in a poset, where the ranks in the
intersection meet semilattice LA are determined by the codimension of elements.
Combining the outputs of our rank function with the Möbius function allows us to
calculate that β0(MA)= 1, β1(MA)= 4, and β2(MA)= 5:

i7 : RLA = rank LA
o7 = {{ideal 0}, {ideal(x+y), ideal(x), ideal(x-y), ideal(y+1)},

{ideal(y,x), ideal(y+1,x-1),ideal(y+1,x), ideal(y+1,x+1)}}
i8 : MF = moebiusFunction LA;
i9 : apply(RLA, r -> sum(r, x -> abs MF#(ideal 0_R, x)))
o9 = {1, 4, 5}

LCM-LATTICES. Let R = K [x1, . . . , xt ] be the polynomial ring in t variables
over the field K , where the degree of xi is the standard basis vector ei ∈ Zt . Let
M = (m1, . . . ,mn) be a monomial ideal in R; then we define the lcm-lattice of M ,



12 Cook, Mapes and Whieldon :::: Partially ordered sets in Macaulay2

1

a3b2c b2c2d abc2d a2cd a3b2d

ab2c2d a2bc2d a3b2cd

a2b2c2d

a3b2c2d

Figure 2. The lcm-lattice for M = (a3b2c, a3b2d, a2cd, abc2d, b2c2d).

denoted by L M , to be the set of all least common multiples of subsets of the gener-
ators of M , partially ordered by divisibility. It is easy to see that L M will always be
a finite atomic lattice. While lcm-lattices are nicely structured, they can be difficult
to compute by hand, especially for large examples or for ideals where L M is not
ranked.

Consider the ideal

M = (a3b2c, a3b2d, a2cd, abc2d, b2c2d)

in R = K [a, b, c, d]. We construct L M in Macaulay2 as follows. See Figure 2 for
the Hasse diagram of L M , as generated by the texPoset method.

i10 : R = QQ[a,b,c,d];
i11 : M = ideal(a^3*b^2*c, a^3*b^2*d, a^2*c*d, a*b*c^2*d, b^2*c^2*d);
i12 : LM = lcmLattice M;

Lcm-lattices, which were introduced by Gasharov, Peeva, and Welker [Gasharov
et al. 1999], have become an important tool used in studying free resolutions of
monomial ideals. There have been a number of results that use the lcm-lattice
to give constructive methods for finding free resolutions for monomial ideals; for
some examples see [Clark 2010; Peeva and Velasco 2011; Velasco 2008].

In particular, Gasharov, Peeva, and Welker [Gasharov et al. 1999] provided a key
connection between the lcm-lattice of a monomial ideal M of R and its minimal
free resolution; namely, one can compute the (multigraded) Betti numbers of R/M
using the lcm-lattice. Let 1(P) denote the order complex of the poset P; then for
i ≥ 1 we have

βi,b(R/M)= dim H̃i−2(1(0̂, b); K )

for all b ∈ L M , and so

βi (R/M)=
∑

b∈L M

dim H̃i−2(1(0̂, b); K ).

These computations can all be done using Posets together with the package
SimplicialComplexes by S. Popescu, G. Smith, and M. Stillman. In particular, we
show that βi,a2b2c2d = 0 for all i with the following calculation:

i13 : D1 = orderComplex(openInterval(LM, 1_R, a^2*b^2*c^2*d));
i14 : prune HH(D1)
o14 = -1 : 0

0 : 0
1 : 0

o14 : GradedModule

Similarly, we show that β1,a3b2cd = 2:
i15 : D2 = orderComplex(openInterval(LM, 1_R, a^3*b^2*c*d));
i16 : prune HH(D2)
o16 = -1 : 0

2
0 : QQ

o16 : GradedModule

HIBI IDEALS. Let P = {p1, . . . , pn} be a finite poset with partial order �, and
let K be a field. The Hibi ideal, introduced by Herzog and Hibi [2005], of P over
K is the squarefree ideal HP in R = K [x1, . . . , xn, y1, . . . , yn] generated by the
monomials

u I :=
∏
pi∈I

xi

∏
pi /∈I

yi ,

where I is an order ideal of P , i.e., for every i ∈ I and p ∈ P , if p � i then p ∈ I .

NB: The Hibi ideal is the ideal of the monomial generators of the Hibi ring, a toric
ring first described by Hibi [1987].

i17 : P = divisorPoset 12;
i18 : HP = hibiIdeal P
o18 = monomialIdeal (x x x x x x , x x x x x y , x x x x y y ,

0 1 2 3 4 5 0 1 2 3 4 5 0 1 2 4 3 5
x x x x y y , x x x y y y , x x x y y y , x x y y y y ,
0 1 2 3 4 5 0 1 3 2 4 5 0 1 2 3 4 5 0 2 1 3 4 5

x x y y y y , x y y y y y , y y y y y y )
0 1 2 3 4 5 0 1 2 3 4 5 0 1 2 3 4 5

Herzog and Hibi [2005] proved that every power of HP has a linear resolution,
and the i-th Betti number βi (R/HP) is the number of intervals of the distributive
lattice L(P) of P isomorphic to the rank-i boolean lattice. Using [Stanley 2012,
Exercise 3.47], we recover this by looking instead at the number of elements of
L(P) that cover exactly i elements:

i19 : betti res HP
0 1 2 3

o19 = total: 1 10 12 3
0: 1 . . .
1: . . . .
2: . . . .
3: . . . .
4: . . . .
5: . 10 12 3

i20 : LP = distributiveLattice P;
i21 : cvrs = partition(last, coveringRelations LP);
i22 : -- Determine the number of elements each element covers.

iCvrs = tally apply(keys cvrs, i -> #cvrs#i);
i23 : -- Turn iCvrs into a list indexed by integers.

gk = prepend(1, apply(sort keys iCvrs, k -> iCvrs#k))
o23 : {1, 6, 3}
i24 : -- Determine the number of intervals of LP isomorphic

-- to boolean lattices of a given rank.
apply(#gk, i -> sum(i..<#gk, j -> binomial(j, i) * gk_j))

o24 : {10, 12, 3}

Moreover, Herzog and Hibi [2005] proved that the projective dimension of HP

is the Dilworth number of P , i.e., the maximum length of an antichain of P .

i25 : pdim module HP == dilworthNumber P
o25 = true

REFERENCES.
[Clark 2010] T. B. P. Clark, “Poset resolutions and lattice-linear monomial ideals”, J. Algebra 323:4
(2010), 899–919. MR 2011b:13037 Zbl 1203.13016

[Gasharov et al. 1999] V. Gasharov, I. Peeva, and V. Welker, “The lcm-lattice in monomial resolu-
tions”, Math. Res. Lett. 6:5-6 (1999), 521–532. MR 2001e:13018 Zbl 0970.13004

[Grayson and Stillman] D. R. Grayson and M. E. Stillman, “Macaulay2: a software system for
research in algebraic geometry”, Available at http://www.math.uiuc.edu/Macaulay2.

[Herzog and Hibi 2005] J. Herzog and T. Hibi, “Distributive lattices, bipartite graphs and Alexander
duality”, J. Algebraic Combin. 22:3 (2005), 289–302. MR 2006h:06004 Zbl 1090.13017

[Hibi 1987] T. Hibi, “Distributive lattices, affine semigroup rings and algebras with straightening
laws”, pp. 93–109 in Commutative algebra and combinatorics (Kyoto, 1985), edited by M. Nagata
and H. Matsumura, Adv. Stud. Pure Math. 11, North-Holland, Amsterdam, 1987. MR 90b:13024
Zbl 0654.13015

[Orlik and Solomon 1980] P. Orlik and L. Solomon, “Combinatorics and topology of complements
of hyperplanes”, Inventiones Math. 56:2 (1980), 167–189. MR 81e:32015 Zbl 0432.14016

[Peeva and Velasco 2011] I. Peeva and M. Velasco, “Frames and degenerations of monomial resolu-
tions”, Trans. Amer. Math. Soc. 363:4 (2011), 2029–2046. MR 2011k:13021 Zbl 1221.13024

[Stanley 2012] R. P. Stanley, Enumerative combinatorics, I, 2nd ed., Cambridge Studies in Ad-
vanced Mathematics 49, Cambridge University Press, 2012. MR 2868112 Zbl 1247.05003

[Velasco 2008] M. Velasco, “Minimal free resolutions that are not supported by a CW-complex”, J.
Algebra 319:1 (2008), 102–114. MR 2008j:13028 Zbl 1133.13015

http://dx.doi.org/10.1016/j.jalgebra.2009.11.029
http://msp.org/idx/mr/2011b:13037
http://msp.org/idx/zbl/1203.13016
http://dx.doi.org/10.4310/MRL.1999.v6.n5.a5
http://dx.doi.org/10.4310/MRL.1999.v6.n5.a5
http://msp.org/idx/mr/2001e:13018
http://msp.org/idx/zbl/0970.13004
http://www.math.uiuc.edu/Macaulay2
http://www.math.uiuc.edu/Macaulay2
http://dx.doi.org/10.1007/s10801-005-4528-1
http://dx.doi.org/10.1007/s10801-005-4528-1
http://msp.org/idx/mr/2006h:06004
http://msp.org/idx/zbl/1090.13017
http://msp.org/idx/mr/90b:13024
http://msp.org/idx/zbl/0654.13015
http://dx.doi.org/10.1007/BF01392549
http://dx.doi.org/10.1007/BF01392549
http://msp.org/idx/mr/81e:32015
http://msp.org/idx/zbl/0432.14016
http://dx.doi.org/10.1090/S0002-9947-2010-04980-3
http://dx.doi.org/10.1090/S0002-9947-2010-04980-3
http://msp.org/idx/mr/2011k:13021
http://msp.org/idx/zbl/1221.13024
http://msp.org/idx/mr/2868112
http://msp.org/idx/zbl/1247.05003
http://dx.doi.org/10.1016/j.jalgebra.2007.10.011
http://msp.org/idx/mr/2008j:13028
http://msp.org/idx/zbl/1133.13015


Cook, Mapes and Whieldon :::: Partially ordered sets in Macaulay2 15

[Wachs 2007] M. L. Wachs, “Poset topology: tools and applications”, pp. 497–615 in Geometric
combinatorics, edited by E. Miller et al., IAS/Park City Math. Ser. 13, Amer. Math. Soc., Provi-
dence, RI, 2007. MR 2383132 Zbl 1135.06001

[Zaslavsky 1975] T. Zaslavsky, Facing up to arrangements: face-count formulas for partitions of
space by hyperplanes, Mem. Amer. Math. Soc. 154, Amer. Math. Soc., Providence, RI, 1975.
MR 50 #9603 Zbl 0296.50010

RECEIVED: 21 Jan 2014 REVISED: 14 Aug 2014 ACCEPTED: 5 Jun 2015

DAVID COOK II:

dwcook@eiu.edu
Department of Mathematics & Computer Science, Eastern Illinois University, Charleston, IN 46613,
United States

SONJA MAPES:

smapes1@nd.edu
Department of Mathematics, University of Notre Dame, Notre Dame, IN 46556, United States

GWYNETH WHIELDON:

whieldon@hood.edu
Department of Mathematics, Hood College, Frederick, MD 21701, United States

msp

http://msp.org/idx/mr/2383132
http://msp.org/idx/zbl/1135.06001
http://msp.org/idx/mr/50:9603
http://msp.org/idx/zbl/0296.50010
mailto:dwcook@eiu.edu
mailto:smapes1@nd.edu
mailto:whieldon@hood.edu
http://msp.org




JSAG 7 (2015), 17–29 The Journal of Software for
dx.doi.org/10.2140/jsag.2015.7.17 Algebra and Geometry

Free resolutions and modules with a
semisimple Lie group action

FEDERICO GALETTO

ABSTRACT: We introduce the package HighestWeights for Macaulay2. This
package provides tools to study the representation-theoretic structure of free
resolutions and graded modules over a polynomial ring with the action of a
semisimple Lie group. The methods of this package allow users to consider
the free modules in a resolution, or the graded components of a module, as
representations of a semisimple Lie group by means of their weights, and to
obtain their decomposition into highest-weight representations.

1. INTRODUCTION. Let R be a polynomial ring over the complex numbers with
a Zm-grading, and let M be a finitely generated graded R-module. Under mild
assumptions on R, for every degree d ∈ Zm each graded component Md is a finite-
dimensional complex vector space. Next assume that G is a complex Lie group
and that there is a degree-preserving C-linear action of G on R compatible with
multiplication, i.e., such that for every g ∈ G, r1, r2 ∈ R we have g · (r1r2) =

(g · r1)(g · r2). We are interested in those R-modules M with a degree-preserving
C-linear action of G compatible with the module structure, i.e., such that for every
g ∈ G, r ∈ R and m ∈ M we have g · (rm)= (g · r)(g ·m). Notice that each graded
component of such a module M is stable under the action of G.

Examples of such modules can arise naturally. For instance, let X be a finite-
dimensional representation of a complex Lie group G. The symmetric algebra
Sym(X), with the standard grading determined by Sym1(X) ∼= X , is an example
of a polynomial ring with a degree-preserving C-linear action of G. The action of
G also extends to the projective space P(X). If V is a projective variety in P(X)
which is fixed by the action of G, then the affine cone V̂ is an affine variety in X
which is fixed by G. Moreover, V̂ is the zero locus of some radical homogeneous
ideal I in Sym(X), and the quotient ring Sym(X)/I , i.e., the affine coordinate ring
of V̂ , is an example of a Sym(X)-module with a compatible G-action.

MSC2010: primary 13P20; secondary 22E46.
Keywords: equivariant free resolution, irreducible representation, highest weight, algebraic torus,

semisimple Lie group, decomposition algorithm.

Let M be a finitely generated graded R-module with the kind of G-action de-
scribed earlier. Denote by m the maximal ideal generated by the variables in R, and
set V0 = M/mM . The vector space V0 is a finite-dimensional graded representation
of G. Assuming G is linearly reductive, the natural projection M→ V0 admits a
section V0→ M which is compatible with the grading and G-action. This section
extends to an R-linear map d0 : F0 → M , where F0 is defined to be V0 ⊗C R.
By construction, F0 is a graded free R-module with a natural action of G that
commutes with d0; moreover, d0 maps a basis of F0 to a minimal generating set
of M . Notice also that F0/mF0 ∼= V0 as a graded representation of G. Next, let
V1 = ker(d0)/m ker(d0). Using the same ideas as before, let F1 be V1⊗C R (so that
F1/mF1 ∼= V1) and define the map d1 : F1→ F0. Again, d1 will be a map of graded
free R-modules that commutes with the action of G and maps a basis of F1 to a
minimal generating set of ker(d0); in addition, d1 ◦ d0 = 0. Iterating this procedure
constructs a minimal free resolution of M with a built-in action of G; in particular,
the procedure is guaranteed to terminate. Since any two minimal free resolutions
of M are isomorphic as complexes, the action of G transfers to every other minimal
free resolution via the isomorphism with the one just constructed.

Summarizing what we said so far, for every finitely generated graded R-module
M with a degree-preserving G-action compatible with the module structure, the
following occurs:

(1) For every degree d ∈ Zm , the graded component Md is a finite-dimensional
representation of G.

(2) If the complex

0←− M←− F0←− F1←− · · · ←− Fn←− 0

is a minimal free resolution of M as an R-module, then the action of G on M
extends to each Fi and there is an isomorphism Fi ∼= (Fi/mFi )⊗C R of graded
R-modules with a G-action. Each Fi/mFi is a finite-dimensional graded
vector space so, for each degree d ∈ Zm , (Fi/mFi )d is a finite-dimensional
representation of G.

When the group G is semisimple, it is a typical problem to decompose a finite-
dimensional representation into irreducible representations. Moreover each irre-
ducible representation is indexed by a so-called highest weight. The main pur-
pose of HighestWeights is to provide users of Macaulay2 [Grayson and Stillman]
with tools to obtain the highest-weight decomposition of the representations Md

and (Fi/mFi )d introduced above. This purpose is achieved by implementing an
algorithm for propagating weights of tori along equivariant maps introduced in
[Galetto 2015].

This article is organized as follows: the next section details the mathematical
assumptions for using this package, Section 3 presents two examples in detail, and
Section 4 contains some final remarks. We have also included a brief review of the
representation theory of semisimple Lie groups in the Appendix.

2. PACKAGE ASSUMPTIONS. Before presenting some examples, we discuss some
assumptions of this package. Using the notation of the introduction, the poly-
nomial ring R must be positive Zm-graded for some positive integer m, in the
sense of [Kreuzer and Robbiano 2005, Definition 4.2.4]. More explicitly, if R =
C[x1, . . . , xn], then R is graded by elements of Zm in such a way that:

(1) each nonzero constant in R has degree 0 ∈ Zm ;

(2) the degree of each variable xi ∈ R is a nonzero vector in Zm , and its first
nonzero entry is positive;

(3) the matrix with rows given by the degrees of the variables x1, . . . , xn has
rank m.

This ensures that, for every finitely generated graded R-module M and for every
degree d ∈ Zm , each graded component Md is a finite-dimensional complex vector
space. The variables in the polynomial ring R must be weight vectors for the action
of the chosen maximal torus in G; this can always be achieved up to a linear change
of variables in R. The user is expected to provide the weight of each variable. Any
monomial ordering on the monomials of R is allowed. All free R-modules must
be endowed with a term over position up/down or position up/down over term
ordering; in Macaulay2 this is established with the declaration of the ring. The
default ordering, term over position up, is fine for most computations, unless the
user needs a different one.

To obtain the decomposition of a graded component Md , the user is expected to
provide a presentation ϕ : F1→ F0 of M in the form of a matrix written with respect
to a homogeneous basis {e1, . . . , er } of F0 such that the residue classes ē1, . . . , ēr

modulo mF0 form a basis of weight vectors of F0/mF0; the user will also need to
provide a list with the weights of ē1, . . . , ēr . For modules with a compatible group
action, presentations of this kind are, in our experience, the most natural. As for
resolutions, the user must provide a list of weights for a basis e1, . . . , er as before,
for any one of the modules Fi . When M = R/I , for a G-stable ideal I in R, the
module F0 is simply R with a trivial G-action; in this case, the user does not need
to input any weight (other than those of the variables of R).

The package WeylGroups, which is loaded automatically by this package, is
used to declare the type of a semisimple group and to handle many weight-related
operations behind the scenes. However, for the purpose of this package, weights

are to be provided simply as lists of integers, not as objects of type Weight as in
WeylGroups.

3. EXAMPLES. For an explanation of the notations and conventions relating to
weights that appear in the following examples, we refer the reader to Section A2.

3.1. The coordinate ring of the Grassmannian. Let E = C6, the standard repre-
sentation of SL6(C), with coordinate basis {e0, . . . , e5}. The Grassmannian V =
Gr(2, E∗) is the projective variety which parametrizes two-dimensional subspaces
of E∗; it is embedded in P(

∧2 E∗) using the Plücker equations [Shafarevich 1994,
Chapter I, §4.1]. Consider

∧2 E∗ as a complex affine space. Let C be the affine cone
over V , i.e., the subvariety of

∧2 E∗ which is the union of all the one-dimensional
subspaces of

∧2 E∗ belonging to V . The space
∧2 E∗ has a natural action of SL6(C)

which fixes C .
Our polynomial ring R is the ring of polynomial functions over

∧2 E∗, i.e., the
symmetric algebra Sym(

∧2 E). The elements pi, j = ei ∧ e j for 0 6 i < j 6 5
form a basis of weight vectors of

∧2 E and will be the variables in R. The defining
ideal of C is generated by the Plücker equations; this ideal, which we call I , can
be conveniently obtained in Macaulay2 using the command Grassmannian. We
resolve the quotient R modulo I as an R-module and call RI the minimal free
resolution:
i1 : printWidth=72; truncateOutput 200;

i3 : I=Grassmannian(1,5,CoefficientRing=>QQ); R=ring I;

o3 : Ideal of QQ[p , p , p , p , p , p , p , p , p , ... ]
0,1 0,2 1,2 0,3 1,3 2,3 0,4 1,4 2,4

i5 : RI=res I; betti RI

0 1 2 3 4 5 6
o6 = total: 1 15 35 42 35 15 1

0: 1 . . . . . .
1: . 15 35 21 . . .
2: . . . 21 35 15 .
3: . . . . . . 1

o6 : BettiTally

Now we load the package and assign weights to the variables of R. First we
input the weights of e0, . . . , e5 in a list L.
i7 : loadPackage "HighestWeights";

i8 : L={{1,0,0,0,0},{-1,1,0,0,0},{0,-1,1,0,0},{0,0,-1,1,0},{0,0,0,-1,1},
{0,0,0,0,-1}};

The weight of pi, j = ei ∧ e j is equal to the sum of the weights of ei and
e j (see Appendix A2). The subscripts of the variables pi, j are the elements of
subsets({0,1,2,3,4,5},2), the 2-subsets of the set {0, 1, 2, 3, 4, 5}. Hence

taking sums of pairs of weights in L over this indexing set will give us a complete
list of weights for the variables pi, j , as listed by Macaulay2:
i9 : W=apply(subsets({0,1,2,3,4,5},2),s->L_(s_0)+L_(s_1))

o9 = {{0, 1, 0, 0, 0}, {1, -1, 1, 0, 0}, {-1, 0, 1, 0, 0}, {1, 0, -1, 1, 0},
-----------------------------------------------------------------------
{-1, 1, -1, 1, 0}, {0, -1, 0, 1, 0}, {1, 0, 0, -1, 1}, {-1, 1, 0, -1, 1},
-----------------------------------------------------------------------
{0, -1, 1, -1, 1}, {0, 0, -1, 0, 1}, {1, 0, 0, 0, ...

o9 : List

We declare D to be the Dynkin type A5, which is the type of the group SL6(C).
We then attach the weights in W to the variables in R with the command setWeights;
the arguments are the ring, the type and the weights of the variables, respectively.
The output will be the highest-weight decomposition of the C-linear subspace
of R generated by its variables; it is given in the form of a Tally, with keys
describing the highest weights of the irreducible representation appearing in the
decomposition and values equal to the multiplicities of those representations. In
this case, we get simply {0, 1, 0, 0, 0} => 1, which means that the decompo-
sition contains only one copy of the irreducible representation with highest weight
{0, 1, 0, 0, 0}, i.e.,

∧2 E , as expected:
i10 : D=dynkinType{{"A",5}}; setWeights(R,D,W)

o11 = Tally{{0, 1, 0, 0, 0} => 1}

o11 : Tally

All monomials in R are weight vectors. To recover the weight of a monomial,
use the command getWeights with the monomial as the argument:
i12 : getWeights(p_(0,1)*p_(1,2))

o12 = {-1, 1, 1, 0, 0}

o12 : List

We can now issue the command highestWeightsDecomposition to obtain
the decomposition of the representations corresponding to the free modules in the
resolution; the only argument is the resolution RI. Suppose the free modules in
RI are F0, . . . , F6. The outermost HashTable in the output has keys equal to the
subscripts of the free modules in RI. The value corresponding to a key i is itself
a HashTable with keys equal to the degrees of the generators of Fi . Finally the
value corresponding to a certain degree d is a Tally containing the highest-weight
decomposition of the representation (Fi/mFi )d , as described earlier:
i13 : highestWeightsDecomposition(RI)

o13 = HashTable{0 => HashTable{{0} => Tally{{0, 0, 0, 0, 0} => 1}}}
1 => HashTable{{2} => Tally{{0, 0, 0, 1, 0} => 1}}
2 => HashTable{{3} => Tally{{1, 0, 0, 0, 1} => 1}}

3 => HashTable{{4} => Tally{{2, 0, 0, 0, 0} => 1}}
{5} => Tally{{0, 0, 0, 0, 2} => 1}

4 => HashTable{{6} => Tally{{1, 0, 0, 0, 1} => 1}}
5 => HashTable{{7} => Tally{{0, 1, 0, 0, 0} => 1}}
6 => HashTable{{9} => Tally{{0, 0, 0, 0, 0} => 1}}

o13 : HashTable

By analyzing this output, we obtain the following description for RI:

R←
∧4 E ⊗ R(−2)← S2,1,1,1,1 E ⊗ R(−3)

← S2 E ⊗ R(−4)⊕S2,2,2,2,2 E ⊗ R(−5)← S2,1,1,1,1 E ⊗ R(−6)

←
∧2 E ⊗ R(−7)← R(−9)← 0.

Next we turn to the coordinate ring of C , i.e., the quotient ring Q = R/I . We
decompose its graded components in the range of degrees from 0 to 4, again with
the command highestWeightsDecomposition. This time the arguments are the
ring followed by the lowest and highest degrees in the range to be decomposed:
i14 : Q=R/I; highestWeightsDecomposition(Q,0,4)

o15 = HashTable{0 => Tally{{0, 0, 0, 0, 0} => 1}}
1 => Tally{{0, 1, 0, 0, 0} => 1}
2 => Tally{{0, 2, 0, 0, 0} => 1}
3 => Tally{{0, 3, 0, 0, 0} => 1}
4 => Tally{{0, 4, 0, 0, 0} => 1}

o15 : HashTable

We deduce that (R/I )d = Sd,d E for d ∈ {0, . . . , 4}. We can also decompose the
graded components of the ring R in a range of degrees or in a single degree:
i16 : highestWeightsDecomposition(R,2)

o16 = Tally{{0, 0, 0, 1, 0} => 1}
{0, 2, 0, 0, 0} => 1

o16 : Tally

For example, R2 =
∧4 E ⊕ S2,2 E . Since the representation

∧4 E appears in R2

but not in (R/I )2, we deduce that it must be in I2, the graded component of I of
degree 2. This can be verified directly by decomposing I2 as follows:
i17 : highestWeightsDecomposition(I,2)

o17 = Tally{{0, 0, 0, 1, 0} => 1}

o17 : Tally

3.2. The Buchsbaum–Rim complex. Let E=C6 with coordinate basis {e1, . . . ,e6}

and F = C3 with coordinate basis { f1, f2, f3}. Denote by R the symmetric algebra
Sym(E ⊗ F); R is a polynomial ring with variables xi, j = ei ⊗ f j . We take M
to be the cokernel of a generic 3× 6 matrix of variables in R. The minimal free
resolution of M is an example of a Buchsbaum–Rim complex [Eisenbud 1995,
Appendix A2.6]. We call this complex BR.

i1 : printWidth = 72; truncateOutput 200;

i3 : R=QQ[x_(1,1)..x_(6,3)];

i4 : G=genericMatrix(R,3,6)

o4 = | x_(1,1) x_(2,1) x_(3,1) x_(4,1) x_(5,1) x_(6,1) |
| x_(1,2) x_(2,2) x_(3,2) x_(4,2) x_(5,2) x_(6,2) |
| x_(1,3) x_(2,3) x_(3,3) x_(4,3) x_(5,3) x_(6,3) |

3 6
o4 : Matrix R <--- R

i5 : M=coker G; BR=res M; betti BR

0 1 2 3 4
o7 = total: 3 6 15 18 6

0: 3 6 . . .
1: . . . . .
2: . . 15 18 6

o7 : BettiTally

The ring R carries a degree compatible action of SL6(C)× SL3(C). Define the
map of graded free R-modules

ϕ : E ⊗ R(−1)→ F∗⊗ R, ei ⊗ 1 7→
3∑

j=1
f ∗j ⊗ xi, j ,

where { f ∗1 , f ∗2 , f ∗3 } is the dual basis in F∗. The matrix of ϕ with respect to the
bases {e1⊗1, . . . , e6⊗1} and { f ∗1 ⊗1, f ∗2 ⊗1, f ∗3 ⊗1} is precisely the generic matrix
G introduced above. Moreover, ϕ is SL6(C)× SL3(C)-equivariant, meaning that
for all g ∈ SL6(C)×SL3(C), e ∈ E and r ∈ R we have ϕ(g · (e⊗r))= g ·ϕ(e⊗r).
This makes its cokernel M a module with a compatible SL6(C)×SL3(C)-action.

The weight of xi, j = ei ⊗ f j is obtained by concatenating the weight of ei with
that of f j . First we record the weights of e1, . . . , e6 in a list e and those of f1, f2, f3

in a list f. Then we concatenate them as illustrated below and attach the resulting
list to the variables xi, j . Care must be taken that the order of the weights matches
the order of the variables:

i8 : loadPackage "HighestWeights";

i9 : e={{1,0,0,0,0},{-1,1,0,0,0},{0,-1,1,0,0}, {0,0,-1,1,0},{0,0,0,-1,1},
{0,0,0,0,-1}};

i10 : f={{1,0},{-1,1},{0,-1}};

i11 : W=flatten table(e,f,(u,v)->u|v)

o11 = {{1, 0, 0, 0, 0, 1, 0}, {1, 0, 0, 0, 0, -1, 1}, {1, 0, 0, 0, 0, 0, -1},
------------------------------------------------------------------------
{-1, 1, 0, 0, 0, 1, 0}, {-1, 1, 0, 0, 0, -1, 1}, {-1, 1, 0, 0, 0, 0, -1},
------------------------------------------------------------------------
{0, -1, 1, 0, 0, 1, 0}, {0, -1, 1, 0, 0, -1, 1}, { ...

o11 : List
i12 : D=dynkinType{{"A",5},{"A",2}}; setWeights(R,D,W)

o13 = Tally{{1, 0, 0, 0, 0, 1, 0} => 1}

o13 : Tally

In order to decompose the representations in a resolution, we need to ensure that
the coordinate basis for at least one of the free modules in the resolution is a basis
of weight vectors, and then we need to input the weights of the elements of that
basis. For our resolution BR, we could choose the first or the second free module.
In fact, the first differential of BR is the map ϕ : E ⊗ R(−1)→ F∗ ⊗ R whose
matrix was written with respect to the bases of weight vectors {e1⊗ 1, . . . , e6⊗ 1}
and { f ∗1 ⊗ 1, f ∗2 ⊗ 1, f ∗3 ⊗ 1}. We choose to work with the first module, i.e., the
codomain of ϕ. Notice that the element 1 ∈ R appearing in the tensor product has
weight zero; hence it does not contribute to the weight of the basis elements. Also
the SL6(C) factor of our group acts trivially on F∗, and hence to obtain the weight
of f ∗1 ⊗ 1 we concatenate {0,0,0,0,0}, the weight of the trivial representation of
SL6(C), with {-1,0}, the weight of f ∗1 . We proceed similarly for the other basis
vectors and record the weights in the list U0:
i14 : U0={{0,0,0,0,0,-1,0},{0,0,0,0,0,1,-1},{0,0,0,0,0,0,1}};

At this point we are ready to decompose BR. To do so, we issue the command
highestWeightsDecomposition with three arguments: the first is BR, the second
is an integer i informing Macaulay2 that we wish to provide the weights in the i-th
free module of the complex, and the third is the list of weights in the coordinate
basis of the i-th module (remember the indexing of the modules starts from zero
in Macaulay2):
i15 : H0=highestWeightsDecomposition(BR,0,U0)

o15 = HashTable{0 => HashTable{{0} => Tally{{0, 0, 0, 0, 0, 0, 1} => 1}}}
1 => HashTable{{1} => Tally{{1, 0, 0, 0, 0, 0, 0} => 1}}
2 => HashTable{{4} => Tally{{0, 0, 0, 1, 0, 0, 0} => 1}}
3 => HashTable{{5} => Tally{{0, 0, 0, 0, 1, 1, 0} => 1}}
4 => HashTable{{6} => Tally{{0, 0, 0, 0, 0, 2, 0} => 1}}

o15 : HashTable

We deduce that BR decomposes as

F∗⊗R←−E⊗R(−1)←−
∧4 E⊗R(−4)←−

∧5 E⊗F⊗R(−5)←−S2 F⊗R(−6)←−0.

If we choose to start from the second module, we need to provide the list of
weights of the elements e1⊗ 1, . . . , e6⊗ 1. The commands are:
i16 : U1={{1,0,0,0,0,0,0},{-1,1,0,0,0,0,0},{0,-1,1,0,0,0,0},{0,0,-1,1,0,0,0},

{0,0,0,-1,1,0,0},{0,0,0,0,-1,0,0}};

i17 : H1=highestWeightsDecomposition(BR,1,U1); H0===H1

o18 = true

Indeed, the decomposition is the same.

As with rings and ideals, we can decompose the graded components of a module.
The difference is that we need to provide a list of weights for the generators of the
presentation used to define the module. For our module M , this is exactly the list
U0 introduced earlier. As usual, we may decompose a single degree or a range.

i19 : highestWeightsDecomposition(M,-1,2,U0)

o19 = HashTable{-1 => Tally{} }
0 => Tally{{0, 0, 0, 0, 0, 0, 1} => 1}
1 => Tally{{1, 0, 0, 0, 0, 1, 1} => 1}
2 => Tally{{0, 1, 0, 0, 0, 0, 2} => 1}

{2, 0, 0, 0, 0, 2, 1} => 1

o19 : HashTable

Since M is generated in degree zero, we see that the output contains an empty
decomposition in degree −1. Whereas we see, for example, that

M2 =
∧2 E ⊗S2,2 F ⊕S2 E ⊗S3,1 F.

4. CLOSING REMARKS. Here we comment on a few points of (potential) interest.

• The method highestWeightsDecomposition provides the main functionality
of this package. This method relies on the method propagateWeights and the
function decomposeWeightsList, both of which are also exported. The method
propagateWeights implements (with minor changes) an algorithm of [Galetto
2015]. The function decomposeWeightsList implements a modified version
of Freudenthal’s multiplicity formula using the algorithm discussed in [de Graaf
2000, Chapter 8.9] and [Moody and Patera 1982]. We do not anticipate the user
employing these commands directly, but they are available for those who wish to
experiment with them. More details are available in the package documentation.

• Multigradings are supported as well as single gradings, whenever they are com-
patible with the group action. An example is included in the documentation that
involves multigradings.

• Decomposing graded components of rings and modules tends to work better in
low degrees, as the dimension of graded components can grow fast.

• All the examples presented in this article and in the documentation of this pack-
age are over the field C of complex numbers for representation-theoretic reasons;
however, all computations are performed in Macaulay2 over the field Q of ratio-
nal numbers. The reader interested in an explanation of why the computational
results obtained over Q can be interpreted over C may consult [Galetto 2015].

• For further concrete examples where the package HighestWeights may be useful,
including links to Macaulay2 files, see [Galetto 2014]. The Macaulay2 package
PieriMaps [Sam 2009] provides means to construct additional examples.

APPENDIX: SOME REPRESENTATION THEORY. We present here a brief review
of the representation theory of semisimple Lie groups with the terminology used
in this paper and by the package HighestWeights. For more details the reader can
consult [Humphreys 1975; 1978; Fulton and Harris 1991]. The manual of the
software package LiE [van Leeuwen et al. 1992] also contains a brief, and in our
opinion well-written, review of the theory.

All Lie groups are intended to be complex Lie groups.

A1. Type and representations of a semisimple Lie group. A (simply connected)
simple Lie group is one of the following: the special linear group SLn+1(C) (type
An), the spin group Spin2n+1(C) (type Bn) or Spin2n(C) (type Dn), the symplectic
group Sp2n(C) (type Cn), or one of the exceptional groups of type E6, E7, E8,
F4 and G2. Every semisimple Lie group G is the quotient of a finite product
of the simple Lie groups above, called the simple components of G, by a finite
subgroup. The type of G is then obtained by concatenating the types of the simple
components.

Let C× be the multiplicative group of nonzero complex numbers. A torus is
a Lie group which is isomorphic to (C×)n , for some positive integer n called
the rank of the torus. Every semisimple Lie group G contains a maximal torus.
All maximal tori are conjugate and hence have the same rank; the rank of G is
defined to be the rank of a maximal torus. Usually one particular maximal torus
is fixed, and it is denoted by T . The character group of T , denoted by X (T ), is
the set of all Lie group homomorphisms χ : T → C×. For any finite-dimensional
representation V of G, there is a unique decomposition V =

⊕
χ∈X (T ) Vχ , where

Vχ = {v ∈ V | ∀t ∈ T, t · v = χ(t)v}. The characters χ such that Vχ 6= 0 are called
weights of V , and dim Vχ is called the multiplicity of χ in V . Each subspace Vχ
is called a weight space, and its nonzero elements are called weight vectors with
weight χ . The weights of V along with their multiplicities uniquely determine V
as a representation of G. Moreover, if T has rank n, there is a group isomorphism
X (T )∼= Zn , and hence weights can be simply recorded as lists of integers.

Every semisimple Lie group G contains a maximal connected solvable subgroup
B, called Borel subgroup, which contains the fixed maximal torus T . Let V be a
finite-dimensional representation of G and let v ∈ V be a weight vector of weight
ω. If v spans a B-stable one-dimensional subspace of V , then v is said to be a
highest-weight vector. The representation V is irreducible if and only if v is, up
to multiplication by a scalar, the only highest-weight vector in V . In this case, V
is the unique irreducible representation of G with highest weight ω, and it is often
denoted by V (ω).

A simple Lie group S of rank n has n fundamental representations. The con-
struction of fundamental representations for each type is detailed in [Fulton and

Harris 1991], and a summary can be found in [Tits 1967]. The concept of fun-
damental representations can be extended to any semisimple Lie group G: if G
has simple components S1, . . . , Sr , then for each component Si and for each fun-
damental representation Vi, j of Si , take Vi, j to be a fundamental representation of
G with a trivial action of all the other components. Fundamental representations
are irreducible, and their highest weights are called fundamental weights. The
fundamental weights are often denoted ω1, . . . , ωn; they form a basis of X (T ) as
a free abelian group, and the isomorphism X (T ) ∼= Zn is usually taken to send
them to the coordinate basis of Zn . From now on we will always write weights as
elements of the additive group Zn , and always in the basis of fundamental weights.

Since the fundamental weights ω1, . . . , ωn form a basis of X (T ), every weight
ω ∈ X (T ) can be written as ω =

∑n
i=1 miωi for some m1, . . . ,mn ∈ Z. Computa-

tionally the weight ω is represented by the list of integers (m1, . . . ,mn). A weight
ω =

∑n
i=1 miωi is called dominant if mi > 0 for all i ∈ {1, . . . , n}. There is a

bijection between dominant weights of X (T ) and irreducible representations of G,
which sends ω to the highest-weight representation V (ω). The identity element of
X (T ), i.e., the zero element of Zn represented by the list (0, . . . , 0), is a dominant
weight, and it corresponds to the trivial representation of G.

A2. The special linear group. The special linear group SLn+1(C) is the set of
complex matrices with determinant 1; it is a simple Lie group of type An . The
subset of diagonal matrices in SLn+1(C) forms a subgroup which is a maximal
torus of rank n. The corresponding Borel subgroup is the subset of upper-triangular
matrices in SLn+1(C).

The fundamental representations of SLn+1(C) are (in order) the exterior powers
Cn+1,

∧2
Cn+1, . . . ,

∧n
Cn+1. Their weights are the fundamental weightsω1, . . . ,ωn ,

and ωi is represented by a list of zeroes with a 1 in the i-th entry.
The irreducible representations of SLn+1(C) are given by the Schur modules

SλCn+1, where λ is a partition with at most n parts. For the construction of Schur
modules the reader may consult [Fulton 1997, Chapter 8] or [Fulton and Harris
1991, Chapter 6]. The representation SλCn+1 is the highest-weight representation
V (ω) for the weight ω = (λ1− λ2)ω1+ · · ·+ (λn−1− λn)ωn−1+ λnωn .

To fix an example, consider the group SL4(C) of type A3. The type of our
group may be input by typing DynkinType{{"A",3}}. The representation C4,
also known as standard representation, has highest weight {1,0,0}, being the
same as the Schur module S1C4. The coordinate basis {e1, e2, e3, e4} of C4 is a
basis of weight vectors, and the weights of e1, e2, e3, e4 are {1,0,0}, {-1,1,0},
{0,-1,1} and {0,0,-1} respectively. If we tensor C4 with itself, we obtain a
new representation of SL4(C) with basis ei ⊗ e j for i, j ∈ {1, 2, 3, 4}. Moreover,
the vector ei ⊗ e j is a weight vector with weight equal to the weight of ei plus

the weight of e j . The same principle of adding weights applies to symmetric and
antisymmetric tensors, as well as to the tensor product of two different representa-
tions of the same group. For example, the weight of e1 ∧ e2 in

∧2
C4 is {0,1,0}.

Indeed,
∧2

C4 is a highest-weight representation with highest weight {0,1,0}, thus
corresponding to the Schur module S1,1C4, and e1 ∧ e2 is its highest-weight vector.
Let us also mention the dual representation (C4)∗; the elements of the dual basis
e∗1, e∗2, e∗3, e∗4 are weight vectors with weights {-1,0,0}, {1,-1,0}, {0,1,-1}
and {0,0,1}. Note that the weight of e∗i is the additive inverse of the weight of ei .
Here the only highest weight is {0,0,1}, so (C4)∗ is isomorphic to

∧3
C4.

When dealing with a (quotient of a) product of groups, say SL4(C)× SL6(C),
we declare the type, listing in order the types of the simple components like this:
DynkinType{{"A",3},{"A",5}}. The representations of SL4(C)× SL6(C) are
obtained by tensoring a representation of SL4(C) with one of SL6(C). Their lists
of weights are then concatenated to form a single list. For example, the repre-
sentation S6,3,1C4

⊗
∧2

C6 has highest weight {3,2,1,0,1,0,0,0}, because the
highest weights of S6,3,1C4 and

∧2
C6 are {3,2,1} and {0,1,0,0,0} respectively.

Similarly, the irreducible representation with highest weight {0,0,0,1,1,1,1,0}
is the tensor product of the irreducible representations C of SL4(C) and S4,3,2,1C6

of SL4(C), which is simply isomorphic to S4,3,2,1C6.

Computing characteristic classes
and the topological Euler characteristic

of complex projective schemes

CHRISTINE JOST

ABSTRACT: The Macaulay2 package CharacteristicClasses provides commands
for the computation of the topological Euler characteristic, the degrees of the
Chern classes and the degrees of the Segre classes of a closed subscheme of
complex projective space. The computations can be done both symbolically
and numerically, the latter using an interface to Bertini. We provide some back-
ground for the implementation, show how to use the package with the help of
examples and compare its performance to other implementations.

INTRODUCTION. The Macaulay2 [Grayson and Stillman] package Characteris-
ticClasses computes degrees of Chern and Segre classes of complex projective
schemes. It also provides a command for computing the topological Euler char-
acteristic. Recall that the Chern classes of a smooth scheme are defined to be the
Chern classes of the tangent bundle. Moreover, the Segre classes of a scheme X
embedded in Pn are defined to be the Segre classes of the normal cone CX Pn of the
embedding. Note that the degree of a d-dimensional cycle class, i.e., the weighted
sum of classes of d-dimensional subvarieties, is defined to be the weighted sum of
the degrees of the subvarieties.

The computations done by the package CharacteristicClasses are based on a
number of related algorithms which reduce the problem to the computation of the
degrees of residuals. Briefly stated, residuals appear in the intersection of hyper-
surfaces containing a certain scheme. The intersection consists of this scheme and
possibly other components, the residual. The algorithm computing Chern classes
is described in [Di Rocco et al. 2011], the one computing Segre classes for possibly
singular schemes in [Eklund et al. 2013], and the computation of the topological
Euler characteristic is described in [Jost 2013]. The residuals can be computed
either symbolically or numerically, which yields symbolic and numeric versions of

The work described in this article was performed while the author was at the University of Stockholm.
MSC2010: primary 14C17; secondary 14Q15, 65H10.

the algorithms. Both versions are implemented in the package, the numeric version
via an interface to Bertini [Bates et al. 2013]. The algorithm is probabilistic, so
with a small but nonzero probability the wrong result is obtained; see the next
section or the documentation for more details.

There are other ways to compute the topological Euler characteristic and degrees
of characteristic classes. The Macaulay2 command euler computes the topolog-
ical Euler characteristic of smooth projective varieties by computing the Hodge
numbers. Aluffi [2003] described algorithms for the computation of degrees of
characteristic classes and the topological Euler characteristic. These algorithms
are implemented in Aluffi’s package CSM, which is not part of Macaulay2 but
can be obtained at www.math.fsu.edu/∼aluffi/CSM/CSM.html. The three afore-
mentioned implementations complement each other, as is shown in [Eklund et al.
2013; Jost 2013] and shortly in the last section of this article. A new promis-
ing algorithm is described by Helmer [2015]. Helmer’s algorithm is included
in version 1.1 of CharacteristicClasses. This article describes version 1.0 of the
package.

COMPUTING DEGREES OF CHERN AND SEGRE CLASSES. We describe how to
make Chern and Segre classes of projective schemes computationally tractable by
computing their degrees. The standard reference for the notions used in this section
(Chern classes, Segre classes and Chow groups) is [Fulton 1984].

In this section, all schemes are defined over the field of complex numbers. Let
X be a k-dimensional closed subscheme of Pn , embedded by i : X ↪→ Pn , and
denote its Chow group by A∗(X) =

⊕k
d=0 Ad(X). Characteristic classes of X

are elements of the graded group A∗(X), whose generators are generally hard to
compute. Hence algorithms for the computation of characteristic classes focus on
computing a coarser invariant, the degrees of the classes. Let α =

∑
i αi [Vi ] be

a cycle class in Ad(X), given as the weighted sum of classes of d-dimensional
subvarieties Vi . Then the degree deg(α) of α is defined to be the weighted sum∑

i αi deg(Vi ) of the degrees of the varieties Vi , seen as subvarieties of Pn . One
can also consider the pushforward of the cycle α to the Chow group of Pn , which is
known to be A∗(Pn)=Z[H ]/(H n+1), where H is the class of a general hyperplane.
The two viewpoints are equivalent because i∗(α)= deg(α)H n−d .

The Chern classes c1(X), . . . , ck(X) of a smooth scheme X are by definition
the Chern classes c1(TX ), . . . , ck(TX ) of the tangent bundle of X . The total Chern
class is defined to be the sum 1+ c1(X)+ · · · + ck(X). Furthermore, the Segre
classes s1(X,Pn), . . . , sk(X,Pn) of a possibly singular X are the Segre classes
s1(CX Pn), . . . , sk(CX Pn) of the normal cone CX Pn of X in Pn . If the embedding
is regular, the normal cone is a vector bundle, called the normal bundle. The total
Segre class is the sum 1+ s1(X,Pn)+ · · ·+ sk(X,Pn).

http://www.math.fsu.edu/~aluffi/CSM/CSM.html

CharacteristicClasses provides the commands chernClass and segreClass,
which compute the pushforwards to the Chow group of Pn of the total Chern and
Segre classes of a given closed subscheme X of Pn . This is equivalent to computing
the degrees of the Chern and Segre classes. (To be precise, if the given scheme is
singular, the command chernClass computes the degrees of the so-called Chern–
Fulton class of the scheme, one of many generalizations of Chern classes to singular
schemes. See [Fulton 1984, Example 4.2.6] or [Eklund et al. 2013, Remark 4.2]
for more details on Chern–Fulton classes.) The closed subscheme X is given by
the generators of a homogeneous ideal in a polynomial ring. The main idea of the
algorithm is to relate the degrees of the Chern and Segre classes to the degrees of so-
called residuals. One randomly chooses a number of hypersurfaces containing the
scheme X , where the number of hypersurfaces is at least the codimension of X . Ac-
cording to a Bertini-type theorem, with probability 1 they intersect either in only X
or in X and some components of expected codimension, the residual. Its degree can
be computed either symbolically or numerically. Symbolically one computes the
saturation using Gröbner basis techniques. The numeric computations can be done
using software for the numeric solution of polynomial equation systems. Character-
isticClasses implements both the symbolic and numeric versions of the algorithms,
using an interface to Bertini [Bates et al. 2013] for the latter. More details on the
algorithms can be found in [Di Rocco et al. 2011] and [Eklund et al. 2013].

In practice, the probability that the intersection of hypersurfaces yields the cor-
rect residual is large but not 1. Hence the implementation may give the wrong
result with a certain nonzero probability. An example of this phenomenon is
given in the documentation of the package, accessed for instance by viewHelp
"probabilistic algorithm".

COMPUTING THE TOPOLOGICAL EULER CHARACTERISTIC. The topological
Euler characteristic of a complex projective variety is the Euler characteristic of the
underlying topological space with the usual Euclidean topology, i.e., the alternating
sum of the Betti numbers. For the computations with CharacteristicClasses, we
use that the topological Euler characteristic is equal to the degree of the top Chern–
Schwartz–MacPherson class. More generally, the package can also compute the
degrees of all the Chern–Schwartz–MacPherson classes. It uses the algorithm de-
scribed in [Jost 2013], which reduces the computation of the degrees of Chern–
Schwartz–MacPherson classes to the computation of Segre classes of singular sub-
varieties. For a more detailed introduction to Chern–Schwartz–MacPherson classes
we refer to [Aluffi 2003] and [Jost 2013].

USING CHARACTERISTICCLASSES. We present four examples demonstrating
how to use CharacteristicClasses. More detailed information on the commands

provided by this package can be found in its documentation, which is shown by
for example using the command viewHelp CharacteristicClasses.

Example 1. We compute the degrees of the Chern classes of a toy example, the
twisted cubic. We start by loading the package and defining the ideal of the
twisted cubic, which is generated by the 2-by-2 minors of a 2-by-3 matrix. Then
we use the command chernClass to obtain the pushforward of the total Chern
class to the Chow ring of Pn . The twisted cubic Ctw has dimension 1; hence it
only has one Chern class c1(Ctw). The total Chern class of the twisted cubic is
1+ c1(Ctw), and thus its pushforward to the Chow ring of the ambient space P3 is
deg(Ctw)H 2

+ deg(c1(Ctw))H 3, where H is the hyperplane class.
Macaulay2, version 1.6
with packages: ConwayPolynomials, Elimination, IntegralClosure,

LLLBases, PrimaryDecomposition, ReesAlgebra, TangentCone

i1 : loadPackage "CharacteristicClasses";
--loading configuration for package "CharacteristicClasses" from

file .../.Macaulay2/init-CharacteristicClasses.m2

i2 : R = QQ[x,y,z,w];

i3 : twistedCubic = minors(2,matrix{{x,y,z},{y,z,w}})

2 2
o3 = ideal (- y + x*z, - y*z + x*w, - z + y*w)

o3 : Ideal of R

i4 : chernClass twistedCubic

3 2
o4 = 2H + 3H

ZZ[H]
o4 : -----

4
H

So the above computations yield that deg(c1(Ctw)) = 2 and confirms that the
degree of the twisted cubic is deg(Ctw) = 3. As the twisted cubic is smooth, the
degree of the top Chern class equals its Euler characteristic χ(Ctw), which is related
to the genus g(Ctw) by χ = 2− 2g. As deg(c1(Ctw))= χ(Ctw)= 2, this confirms
that the genus is 0 and the twisted cubic is a rational curve.

All computations can also be done numerically using an interface to Bertini.
Version 1.3 or higher of Bertini needs to be installed and the package must be
configured correctly. For more information on the configuration, use viewHelp
"configuring Bertini". Then by using the value Bertini for the option
ResidualStrategy, the computations are done numerically using Bertini, instead
of via Gröbner basis computations.

i5 : chernClass(twistedCubic, ResidualStrategy=>Bertini)

3 2
o5 = 2H + 3H

ZZ[H]
o5 : -----

4
H

Example 2. We proceed with an example of the computation of Segre classes.
The Whitney umbrella is a singular surface in P3. Over the reals, it looks like an
umbrella, and the singular locus like the handle of the umbrella. We compute the
degree of the first Segre class s1(S,Pn) of the singular locus S. Observe that in
the following the singular locus is represented in Macaulay2 not as an ideal, but
as a projective variety. All commands in this package work for both ideals and
projective varieties.

i6 : whitney = ideal(x^2*w - y^2*z);

o6 : Ideal of R

i7 : handle = Proj singularLocus whitney;

i8 : segreClass handle

2
o8 = H

ZZ[H]
o8 : -----

4
H

We see that the pushforward of the Segre class of the handle is 1 · H 2
+ 0 · H 3;

hence deg(s1(S,Pn)), the degree of its first Segre class, is 0. By [Aluffi 1994] the
degree of the first Segre class is deg(s1(S,Pn)) = ν − 2, where ν is the number
of pinch points of the surface. It follows that the Whitney umbrella has two pinch
points. Only one of them is visible in the real patch {w 6= 0} in which the Whitney
umbrella is usually drawn; it is the tip of the umbrella.

Example 3. We use the Whitney umbrella from the above example to demon-
strate the difference between different generalizations of Chern classes of smooth
schemes to singular ones. CharacteristicClasses computes the degrees of two
different generalizations, Chern–Schwartz–MacPherson classes and Chern–Fulton
classes. The commands CSMClass and chernClass compute degrees of Chern–
Schwartz–MacPherson and Chern–Fulton classes, respectively.

i9 : CSMClass whitney

3 2
o9 = 4H + 4H + 3H

ZZ[H]
o9 : -----

4
H

i10 : chernClass whitney

3 2
o10 = 9H + 3H + 3H

ZZ[H]
o10 : -----

4
H

For smooth schemes, the Chern–Schwartz–MacPherson classes and Chern–Fulton
classes agree with the Chern classes. In the singular case, however, the two differ-
ent generalizations disagree, as can be seen for instance from this example.

Example 4. We continue with an example from algebraic statistics, Example 2.2.2
of [Drton et al. 2009], which we will use to demonstrate computations of the topo-
logical Euler characteristic. The theory behind the computations is described in
more detail in [Jost 2013]. The random censoring model with two events is a
statistical model implicitly described by the ideal

(2p0 p1 p2+ p2
1 p2+ p1 p2

2 − p2
0 p12+ p1 p2 p12)

in the polynomial ring C[p0, p1, p2, p12], where p0, p1, p2 and p12 describe the
probabilities of two events to occur before or after a third event. Given experi-
mental data, one would like to compute the values for the probabilities p0, p1,
p2 and p12 which best describe the data, by maximizing the likelihood function.
However, the function may have several stationary points, and methods such as the
Newton method may only find a local maximum. Hence it makes sense to define
the maximum likelihood degree as the number of critical points of the likelihood
function, as was done in [Catanese et al. 2006]. By a theorem of Huh [2013],
for a large class of examples the maximum likelihood degree equals the signed
topological Euler characteristic of a certain open subvariety of the model. In this
case, it is the topological Euler characteristic of the open subvariety

V (2p0 p1 p2+p2
1 p2+p1 p2

2−p2
0 p12+p1 p2 p12)\V (p0 p1 p2 p12(p0+p1+p2+p12)),

where p0 p1 p2 p12 6= 0 means that no probability should be zero, and p0 + p1 +

p2 + p12 6= 0 means that the probabilities should sum up to 1. We compute the

topological Euler characteristic using the command eulerChar together with the
inclusion-exclusion principle.

i11 : S = QQ[p0,p1,p2,p12];

i12 : randomCensoring = ideal(2*p0*p1*p2 + p1^2*p2
+ p1*p2^2 - p0^2*p12 + p1*p2*p12);

o12 : Ideal of S

i13 : boundary = ideal( p0*p1*p2*p12*(p0+p1+p2+p12) )
+ randomCensoring;

o13 : Ideal of S

i14 : eulerChar randomCensoring

o14 = 5

i15 : eulerChar boundary

o15 = 2

It follows that the topological Euler characteristic of

V (2p0 p1 p2+p2
1 p2+p1 p2

2−p2
0 p12+p1 p2 p12)\V (p0 p1 p2 p12(p0+p1+p2+p12))

is 5−2= 3. Hence the maximum likelihood degree of the random censoring model
is 3, which confirms the result in [Drton et al. 2009].

PERFORMANCE. As stated in the introduction, the topological Euler character-
istic can be computed by four different implementations, and degrees of Chern
and Segre classes can be computed by three different implementations. The run-
ning times of these implementations, except Helmer’s 2014 implementation, are
compared in more detail in [Eklund et al. 2013] and [Jost 2013]. Here, we briefly
compare the running times of CSMClass in CharacteristicClasses to CSM in Aluffi’s
implementation CSM (the December 2011 version), Helmer’s CSM in his prototype
Macaulay2 implementation, and the routine euler from Macaulay2. Observe that
the latter only works for nonsingular varieties and takes a projective variety as
input. We use a 1.40 GHz two-core processor with 4 MB RAM. The results are
summarized in Table 1.

All computations were done over the rational numbers. The ideal of the smooth
surface is generated by the 2-by-2 minors of a 2-by-3 matrix of random linear forms.

As Table 1 shows, the Macaulay2 routine euler, the symbolic implementation
by Aluffi and the one from this package complement each other. The performance
of Helmer’s implementation is promising. The numeric implementation in this
package is actually slower than the symbolic implementation for small examples
like those shown in Table 1. However, for very large examples the symbolic

Input CSMclass
(symbolic)

CSMclass
(numeric)

Aluffi’s
CSM

Helmer’s
CSM

euler

twisted cubic < 1 47 2 < 1 < 1
smooth surface in P4

defined by minors
34 2131 14777 3 88

Segre embedding of
P1
×P2 in P5

10 285 1 1 < 1

Table 1. Comparison of run times (all times given in seconds).

methods may not terminate due to insufficient memory; the numeric implemen-
tation does not have this problem.

